AUTOMATED SYSTEM FOR ANALYSIS OF RUNOFF HYDROGRAPHS

March 1966 ARS 41-120
Agricultural Research Service

UNITED STATES DEPARTMENT OF AGRICULTURE

CONTENTS

	Page
Introduction	3
Hydrograph function	3
Function fitting	4
Application	5
Conclusions	10
Symbols used in the program	10-
Appendix I--fortran II source program	11
Appendix II--input formats--hydrograph laboratory program No. 1	14
Appendix III--example problems	15

AUTOMATED SYSTEM FOR ANALYSIS OF RUNOFF HYDROGRAPHS

by D. L. Brakensiek[1]

INTRODUCTION

Hydrologists often utilize dimensionless hydrographs to predict simple hydrographs (single peak with the recession not sustained by rainfall or return flow). The dimensionless hydrograph as introduced in this paper has a discharge scale of (q/q_p) and a time scale of (t/t_p) where

q = runoff rate

q_p = peak runoff rate

t = time

t_p = time to peak rate

By using the assumptions of superposition and linearity, hydrologists develop complex hydrographs from a summation of simple hydrographs.

Hydrographs for simple runoff events have been noted to be similar in shape to skewed statistical frequency functions. Gray[2] and Reich[3] utilized this justification for fitting a gamma function to a number of observed hydrographs. Relating the fitted function parameters to hydrologic and geomorphic variables makes it possible to predict hydrographs for ungaged watersheds.

This study considers a function similar to the gamma function, except that it has only one parameter. Even though the function is a one-parameter function, it is used in a two-parameter fashion, that is, two values of the parameter are estimated, one for the rising and one for the falling side of the hydrograph. A computer program was written to automate the fitting procedure. A program listing and example are presented in the appendix.

The function was fitted to 100 hydrographs and the quality of fit was evaluated from the deviation of the estimated hydrographs from the observed hydrographs.

HYDROGRAPH FUNCTION

As in any instance of curve fitting, only empirical justification can be presented for the fitting function. The function developed for this study is written as

$$q/q_p = \mathrm{EXP}\left(-n\left(\ln(t/t_p) + 2\left(1/\sqrt{t/t_p} - 1\right)\right)\right) \qquad [1]$$

[1] Soil and Water Conservation Research Division, Agricultural Research Service, U.S. Department of Agriculture, at Beltsville, Md.

[2] Gray, Don M. Deviation of Hydrographs for Small Watersheds from Measurable Physical Characteristics. Iowa State Univ. Science and Tech., Dept. Agr. Engin., Res. Bul. 506, pp. 515-570. 1960.

[3] Reich, B. M. Design Hydrographs for Very Small Watersheds from Rainfall. Colo. State Univ., Civil Engin. Sect. Unnumb. Pub. 57 pp. 1962.

or, if we define a transformed time scale as

$$T = \ln (t/t_p) + 2 \ (1\sqrt{t/t_p} -1)$$

then equation 1 is written as

$$q/q_p = EXP \ (-nT) \qquad [2]$$

where n is the parameter to be estimated. Equation 2 is now similar to the usual exponential recession function. The function 1 is fitted to the rising and falling side of the dimensionless hydrograph. Hence, a two-parameter system is fitted. Of course, estimates of q_p and t_p are required to recapture the natural hydrograph.

FUNCTION FITTING

The Fortran II source program is listed in Appendix I. It was written for an IBM 1620[4] computer (40 K memory with indirect addressing). The program performs essentially four types of calculations.

(1) The tabulated time scale (civil or military time) of the natural hydrograph is converted to decimal hours. The decimal hour scale is then converted to an accumulated hourly time scale.

(2) A search is made for the peak rate and time-to-peak rate of the natural hydrograph. The time and discharge scales of the natural hydrograph are then converted to dimensionless scales.

(3) The dimensionless hydrograph scales are transformed in accordance with the fitting function:

(a) Discharge scale $= \ln (q/q_p)$

(b) Time scale $= \ln (t/t_p) + 2 \ (1/\sqrt{t/t_p} -1)$

(4) The rising and falling side parameters are estimated by the usual least squares formula applied to

$$\ln(q/q_p) = -nT.$$

It was apparent from manual fitting trials that a closer fit is more important in high-flow parts of the hydrograph than in the extreme tail parts of the hydrograph. To incorporate this into the fitting program, a cutoff constant was utilized. By varying its value, the portion of the hydrograph contributing to estimation of the parameter is controlled. Values of this constant were from 3.0 in decrements of 0.5 down to 0.5. In terms of cutoff of the (q/q_p) - scale, the equivalence is, for example,

$C = 1.$, $(q/q_p) = 0.367$

$C = 2.$, $(q/q_p) = .135$

$C = 3.$, $(q/q_p) = .049$

Thus, using a cutoff of 2.0, the function is fitted to values of (q/q_p) between 0.135 and 1.00. The program simply selects the closest fit on the basis of minimum sums of squares of observed minus calculated from the sequence of C-values.

[4] Mention of a trade name does not constitute a guaranty or warranty of the product named and does not signify that this product is approved to the exclusion of other comparable products.

The final output of the program provides the best fitting hydrograph, the observed hydrograph, time to peak, peak rate, and the rising side and falling side parameter values.

In Appendix II the data input formats are given. For each hydrograph, one parameter card is required (C is usually valued initially as 3.). One data card is required for each line of data.

Appendix III presents a sample set of input data and the corresponding program output. Experience to date indicates that 20 to 30 hydrographs can be fitted per hour, depending on the number of entries per hydrograph.

APPLICATION

An initial study was undertaken to assess the fitting value of this hydrograph function. The study utilized 100 published hydrographs found in "Selected Runoff Events from Small Agricultural Watersheds in the United States"[5] and "Hydrologic Data for Experimental Agricultural Watersheds in the United States, 1956-59".[6] Table 1 presents the watershed and storm identification together with the estimated values of the parameters.

Table 1.--Listing of watersheds and storms studied and fitted parameters.

Location and watershed	Storm Month-day-year	n Rise	n Fall
Bentonville, Ark.:			
W-5........................	5-25-39	-5.4313	-7.9732
W-5........................	6-9-52	-34.3306	-20.7748
Safford, Ariz.:			
W-I........................	7-19-57	-1.4565	-10.2684
W-II........................	8-20-56	-3.0521	-7.8796
W-II........................	7-16-59	-4.3990	-20.6930
W-V........................	8-28-57	-.6709	-3.3382
W-V........................	8-30-57	-.4415	-2.6094
Tombstone, Ariz.:			
W-2........................	7-20-59	-6.2263	-1.6997
Colorado Springs, Colo.:			
W-IV........................	8-13-45	-33.5999	-22.2823
Americus, Ga.:			
W-IV........................	8-19-42	-39.5155	-14.8924
W-IV........................	1-17-43	-11.8268	-7.0215

[5] U.S. Agricultural Research Service. Selected Runoff Events for Small Agricultural Watersheds in the United States. U.S. Dept. Agr., Agr. Res. Serv., Soil and Water Conservation Research Division, 374 pp. January 1960.

[6] Hobbs, Harold W., comp. Hydrologic Data for Experimental Agricultural Watersheds in the United States, 1956-59. U.S. Dept. Agr. Misc. Pub. No. 945, 672 pp. November 1963.

Table 1.--Listing of watersheds and storms studied and fitted parameters--Continued.

Location and watershed	Storm Month-day-year	n Rise	n Fall
Edwardsville, Ill.:			
W-1........................	5-27-38	-83.8372	-40.3424
W-4........................	5-27-38	-90.0152	-21.7968
Iowa City, Iowa:			
Ralston Creek..............	7-18-56	-90.6247	-10.8163
College Park, Md.:			
W-1........................	7-22-45	-46.3302	-32.6977
W-1........................	6-15-54	-11.1429	-18.2174
W-2........................	7-22-45	-9.3360	-14.5556
W-6........................	8-27-43	-5.2201	-16.5945
W-6........................	7-22-45	-15.0175	-14.3344
W-7........................	8-10-42	-31.7859	-27.0364
W-7........................	7-22-45	-13.4699	-31.3912
Oxford, Miss.:			
W-4........................	4-3-58	-21.2269	-8.2025
W-4........................	9-9-59	-3.1856	-10.1465
W-5........................	4-3-58	-8.0191	-7.8754
W-5........................	6-10-59	-10.1031	-9.7004
W-5........................	6-11-59	-4.3152	-6.4076
W-10.......................	5-22-59	-113.8592	-27.1027
W-28.......................	7-22-58	-25.3347	-15.0764
W-28.......................	9-9-59	-4.1932	-5.9237
W-35.......................	5-22-59	-19.7394	-17.0422
WC-1.......................	5-26-59	-16.9993	-30.4440
WC-2.......................	5-26-59	-10.0764	-19.8004
WC-3.......................	5-26-59	-26.7372	-32.9694
Hastings, Nebr.:			
W-3........................	5-1-57	-45.1365	-12.8659
W-3........................	6-15-57	-13.6296	-18.5796
W-5........................	7-3-59	-9.8426	-12.9281
1-H........................	6-16-57	-31.5879	-9.4677
1-H........................	6-12-58	-5.9608	-10.9489
2-H........................	6-12-58	-8.8988	-14.8207
4-H........................	5-4-59	-11.9878	-5.7904
Albuquerque, N. Mex.:			
W-I........................	9-8-47	-20.4245	-13.0889
W-I........................	8-19-56	-18.2309	-4.6776
W-I........................	8-9-57	-3.4505	-7.7425
W-I........................	8-14-58	-32.3868	-46.9678
W-II.......................	8-24-57	-25.8804	-11.8369
W-III......................	8-19-56	-5.6182	-8.4936

Table 1.--Listing of watersheds and storms studied and fitted parameters--Continued.

Location and watershed	Storm Month-day-year	n Rise	n Fall
Mexican Springs, N. Mex.:			
W-2	9-5-40	-23.3866	-39.2123
W-11	8-26-39	-1.8471	-6.9189
High Point, N.C.:			
West Fork of Deep River	5-2-39	-14.0067	-4.4998
	12-26-38	-6.4381	-20.8938
	2-26-39	-3.2270	-5.6469
Coshocton, Ohio:			
131	6-12-57	-2.7887	-10.3692
127	6-12-57	-184.6239	-19.9397
110	6-12-57	-3.0373	-25.3157
111	6-12-57	-50.3599	-39.2510
187	6-12-57	-59.3227	-11.3100
169	6-12-57	-83.1074	-43.8798
183	8-16-47	-15.9781	-15.1650
196	8-16-47	-313.7573	-97.4145
94	6-12-47	-44.8717	-32.7548
Hamilton, Ohio:			
W-1	5-17-43	-10.0951	-3.3781
W-1	7-7-43	-37.8762	-10.8566
Cherokee, Okla.:			
W-1	6-24-58	-29.3441	-8.7506
W-1	9-25-59	-7.1256	-13.5323
W-1	5-28-60	-14.0688	-42.1781
W-2	6-24-58	-36.5202	-21.4625
W-2	10-13-59	-29.5597	-20.0797
W-3	5-28-60	-8.3606	-23.3567
W-4	6-24-58	-5.3942	-6.3858
W-4	5-28-60	-4.6167	-21.8304
Guthrie, Okla.:			
W-VI	9-8-42	-24.4832	-7.7950
W-VI	6-26-45	-37.9473	-8.0815
Stillwater, Okla.:			
W-1	4-18-57	-93.7714	-341.2761
W-1	6-27-57	-38.4060	-153.5767
W-3	6-27-57	-4.1090	-12.6162
Riesel, Tex.:			
W-1	6-10-41	-33.4406	-19.3325
W-1	3-26-46	-195.0137	-64.3905

Table 1.--Listing of watersheds and storms studied and fitted parameters--Continued.

Location and watershed	Storm Month-day-year	n Rise	n Fall
Riesel, Tex.--Continued:			
W-1	4-27-49	-29.2257	-33.9124
W-1	4-24-57	-262.7649	-117.3924
W-2	6-23-59	-310.2265	-81.1602
W-6	4-24-57	-7.9467	-13.9224
W-6	6-23-59	-63.8263	-22.4897
W-10	4-24-57	-10.4061	-15.1351
W-10	6-23-59	-83.3682	-53.8528
Y-6	4-24-57	-17.9326	-6.6444
Y-6	6-4-57	-6.0614	-5.8565
Y-6	6-23-59	-33.0627	-27.4650
SW-12	6-23-59	-38.7213	-53.1431
SW-17	3-31-57	-24.2796	-6.5723
SW-17	4-24-57	-7.9012	-17.3511
Staunton, Va.:			
Bell Creek			
W-1	9-8-48	-22.5718	-10.8254
Fennimore, Wis.:			
W-1	8-5-51	-84.5015	-99.8869
W-2	8-5-51	-12.0619	-25.4745
W-3	8-12-43	-9.7027	-6.2509
W-3	7-11-44	-455.6838	-10.9080
W-3	6-28-45	-45.4637	-9.3694
W-3	6-24-49	-1.5383	-1.5163
LaCrosse, Wis.:			
CW	8-16-40	-14.1442	-10.9085
CW	6-29-41	-3.0925	-5.1205
CW	9-15-41	-21.7221	-27.1035

The fitting ability of the hydrograph function was evaluated from each of the storms listed in table 1. As mentioned, for each estimated parameter there is a sum of squares of deviation for the difference between the observed and estimated hydrograph points. This was printed out for this evaluation. All of these were grouped for the rising and falling parameters according to magnitude. In table 2 the total frequency distribution of the quality of fit is presented. Reference is made to figures 1 and 2 for quantitative measures of the quality of fit.

EXCELLENT	Figure 1, rising and falling side, is an excellent fit.
GOOD	Figure 2, rising side, is at the lower end of the good fit scale.
FAIR TO POOR	Figure 2, falling side, is a poor fit.

Such hydrographs as figure 2 should really not be fitted, since the rising side is characteristic of an initial runoff period followed by the main runoff period, and the recession side is somewhat sustained by a late occurrence of rainfall or return flow. Some evidence of better fitting ability on the falling limb of the hydrograph is shown in table 2.



Figure 1.--Hydrograph fit.



Figure 2.--Hydrograph fit.

Table 2.--Evaluation of the fitting system.

Quality of fit	Percent of cases	
	Rising side	Falling side
Excellent[1].............	28.0	30.0
Good[2]..................	46.0	53.0
Fair to poor[3]..........	26.0	17.0

[1]Both sides of Figure 1.
[2]Rising side of Figure 2 is at the lower end of the good fit scale.
[3]Falling side of Figure 2 is a poor fit.

CONCLUSIONS

To date no particular difficulties have been encountered in the application of the computer program.

The function presented in this study for fitting dimensionless hydrographs has performed satisfactorily when applied to strictly simple hydrographs. The poor fits often resulted from using complex hydrographs.

The success of this study has made it feasible to examine a much larger population of simple hydrographs. Development of parameter prediction equations can then proceed. Once these are developed, it would be possible to estimate simple hydrographs much in the same fashion as unit hydrographs are estimated.

SYMBOLS USED IN THE PROGRAM

Symbol	Description	Symbol	Description
TIME (I)	Time in decimal hours.	QSCALE (I)	Transformed discharge scale.
TIM	Time increment in decimal hours.	SUM1	Uncorrected sum of cross products.
TIME1 (I)	Accumulated decimal hours.	SUM2	Uncorrected sum of squares.
TPEAK	Time to peak in decimal hours.	S1 and S2	Parameters.
PEAK	Peak discharge.	QC (I)	Calculated discharge scale.
R	Reciprocal of peak discharge.	SSR	Sums of squares of deviations (rising).
T	Reciprocal of time to peak.	SSF	Sums of squares of deviations (falling).
DISCH1 (I)	Dimensionless discharge scale.	DEV	Deviations, observed minus calculated.
TIME2 (I)	Dimensionless time scale.	C	Cut-off constant.
TSCALE (I)	Transformed time scale.		

OPTIONAL SOURCE PROGRAM CHANGE

The Fortran II Source Program as listed in Appendix I may occasionally result in an underflow error message via the typewriter. This does not affect the fitting procedure as the particular hydrograph ordinate, which is set to zero, is actually near zero, i.e., EXP (N) where N < - 227.9. However, one may program around this by making the following source program changes. In place of lines 102-103 inclusive substitute the following:

```
 76      E = S11*TSCALE(I)
         IF(E+200.)775,775,776
775      QC(I) = 0.0
         GO TO 80
776      QC(I) = EXPF(E)
         GO TO 80
 77      E = S22*TSCALE(I)
         IF(E+200.)775,775,777
777      QC(I) = EXPF(E)
```

APPENDIX I

FORTRAN II SOURCE PROGRAM

```
C       HYDROGRAPH LAB PROGRAM NO. 1
C       MATHEMATICAL FIT OF A DIMENSIONLESS HYDROGRAPH
C       ***********************************************************************
C       ***********************************************************************
        DIMENSION TSCALE (100),QSCALE (100),DISCH1(100),QC(100)
        DIMENSION TIME(100),TIME 1(100),TIME 2(100),DISCH(100)
        DIMENSION S1(10),S2(10),SSR(10),SSF(10)
   40   FORMAT (I3,F3.1,3I2,I4)
   41   FORMAT (2F10.4)
   60   FORMAT (3I5,I5)
   62   FORMAT (4F10.4,F12.8)
    2   FORMAT (2F2.0,I2,F7.4)
    3   FORMAT (4F10.4)
    4   FORMAT (2F10.4)
C       ***********************************************************************
   25   READ 40,N,C,YEAR,MO,DAY,WTS
C       ***********************************************************************
C       CLOCK TIME TO HOURS
        DO 103 I=1,N
        READ 2,T1,T2,T3,DISCH(I)
        IF (T3)998,6,5
    5   IF (T1-12.)8,999,8
  999   IF (T2)6,8,6
    8   T1=T1+12.
        GO TO 6
  998   IF (T1-12.)6,997,6
  997   IF (T2)996,6,996
  996   T1=T1-12.
    6   TIME(I)=T1+T2*.016667
        IF (I-1)101,102,101
C       ***********************************************************************
C       TIME ACCUMULATION
  101   TIM=TIME(I)-TIME(I-1)
        IF (TIM)7,7,995
    7   TIM=TIM+24.
        GO TO 995
  102   TIME1(I)=0.0
        GO TO 103
  995   TIME 1(I)=TIME 1(I-1)+TIM
  103   CONTINUE
C       ***********************************************************************
C       PEAK AND TIME TO PEAK
        TPEAK=TIME 1(1)
        PEAK=DISCH(1)
        DO 10 I=2,N
        IF (PEAK-DISCH(I))9,10,10
    9   PEAK=DISCH(I)
        TPEAK=TIME 1(I)
   10   CONTINUE
C       ***********************************************************************
C       OBSERVED DIMENSIONLESS HYDROGRAPH
        R=1./PEAK
        T=1./TPEAK
        DO 15 I=1,N
        DISCH 1(I)=DISCH(I)*R
   15   TIME 2(I) =TIME 1 (I)*T
C       ***********************************************************************
C       TIME AND DISCHARGE SCALE
        IF (TIME1(1))400,400,401
  400   I1=2
        GO TO 601
  401   I1=1
```

FORTRAN II SOURCE PROGRAM--Continued

```
  601 IF (DISCH1(N))500,500,501
  500 M=N-1
      GO TO 402
  501 M=N
  402 DO 18 I=I1,M
      A=LOGF(TIME 2(I))
      B=SQRTF(TIME 2(I))
      B=1./B
      B=(B-1.)*2.
      TSCALE(I)=A+B
   18 QSCALE(I)=LOGF(DISCH 1(I))
   21 CONTINUE
C     ******************************************************************
C     PARAMETER ESTIMATION
      KK1=1
      J1=0
  807 J1=J1+1
      SUM1=0.
      SUM2=0.
      DO 58 I=I1,M
      IF (QSCALE(I)+C)58,51,51
   51 SUM1=TSCALE(I)*QSCALE(I)+SUM1
      SUM2=TSCALE(I)*TSCALE(I)+SUM2
      IF(TSCALE(I)-.000001) 56,58,58
   56 S11=SUM1/SUM2
      SUM1=0.
      SUM2=0.
   58 CONTINUE
      S22=SUM1/SUM2
      S1(J1)=S11
      S2(J1)=S22
C     ******************************************************************
C     CALCULATED DIMENSIONLESS HYDROGRAPH
  900 QC(1)=0.
      J=1
      DO 80 I=I1,M
      IF(TSCALE(I)-.000001) 75,74,74
   75 J=J+1
   74 GO TO (76,77),J
   76 QC(I)=EXPF(S11*TSCALE(I))
      GO TO 80
   77 QC(I)=EXPF(S22*TSCALE(I))
   80 CONTINUE
      IF(N-M)820,820,802
  802 QC(N)=0.
  820 GO TO(801,16),KK1
  801 SSR1=0.
      SSF1=0.
      DO 803 I=1,N
      DEV=DISCH1(I)-QC(I)
      DEV=DEV*DEV
      SSR1=SSR1+DEV
      IF (QC(I)-QC(I+1))803,803,804
  803 CONTINUE
  804 DO 805 K=I,N
      DEV=DISCH1(K)-QC(K)
      DEV=DEV*DEV
      SSF1=SSF1+DEV
  805 CONTINUE
      SSR(J1)=SSR1
      SSF(J1)=SSF1
      C=C-.5
      IF(C)806,806,807
```

APPENDIX I

FORTRAN II SOURCE PROGRAM--Continued

```
806 J1=1
    JR=1
    XR=SSR(J1)
    DO 810 J1=2,6
    IF(SSR(J1)-XR)811,810,810
811 XR=SSR(J1)
    JR=J1
810 CONTINUE
    J1=1
    JF=1
    XF=SSF(J1)
    DO 850 J1=2,6
    IF(SSF(J1)-XF)851,850,850
851 XF=SSF(J1)
     JF=J1
850 CONTINUE
    PUNCH 72
 72 FORMAT (1X,5H YEAR,1X,3H MO,1X,4H DAY,2X,4H WTS)
    PUNCH 60,YEAR,MO,DAY,WTS
    PUNCH 71
 71 FORMAT (3X,5H PEAK,3X,6H TPEAK)
    PUNCH 41,PEAK,TPEAK
    PUNCH 70
 70 FORMAT (5X,5H RISE,5X,5H FALL)
    PUNCH 41,S1(JR) ,S2(JF)
    S11=S1(JR)
    S22=S2(JF)
    KK1=2
    GO TO 900
 16 PUNCH 73
 73 FORMAT (3X,6H HOURS,5X,5H T/TP,5X,2H Q,6X,5H Q/QP,5X,6H QCALC)
 90 PUNCH 62,(TIME1(I),TIME2(I),DISCH(I),DISCH1(I),QC(I),I=1,N)
    GO TO 25
    END
```

APPENDIX II

INPUT FORMATS

HYDROGRAPH LABORATORY PROGRAM NO. 1

Parameter Card

Columns	Format	Symbol	Description
1 - 3	XXXX	N	Number of lines of data (max. = 100)
4 - 6	XX.X	C	Cut-off constant for QSCALE
7 - 8	XX	YEAR	Last two digits of year
9 - 10	XX	MO	Month number (1 - 12)
11 - 12	XX	DAY	Day number (1 - 31)
13 - 16	XXXX	WTS	Black Book watershed number (without decimal)

Data Cards

Columns	Format	Symbol	Description
1 - 2	XX.	T1	Hour digit of time
3 - 4	XX.	T2	Minute digit of time
5 - 6	±X	T3	0, military time -1, A.M. and noon +1, P.M. and midnight
7 - 13	XXX.XXXX		Discharge

Note: The time interval between discharge tabulations must not exceed 24 hours.

APPENDIX III

EXAMPLE PROBLEMS

001 is parameter card   002 to 035 are data cards

```
 34 3 57 524 426                                   001
 251 1    0056                                     002
 252 1    0144                                     003
 256 1    1300                                     004
 258 1    3420                                     005
 301 1    8300                                     006
 304 1   13000                                     007
 307 1   16800                                     008
 310 1   19400                                     009
 312 1   20800                                     010
 315 1   21800                                     011
 317 1   22000                                     012
 319 1   21700                                     013
 321 1   21300                                     014
 324 1   20200                                     015
 327 1   18500                                     016
 330 1   16500                                     017
 336 1   12500                                     018
 340 1   10000                                     019
 345 1    8240                                     020
 355 1    5400                                     021
 401 1    3530                                     022
 405 1    2990                                     023
 410 1    2680                                     024
 422 1    1840                                     025
 439 1    1150                                     026
 455 1    0756                                     027
 515 1    0509                                     028
 650 1    0144                                     029
 800 1    0072                                     030
1159 1    0018                                     031
1200 1    0018                                     032
 100-1    0015                                     033
 200-1    0010                                     034
 300-1    0005                                     035
```

The final hydrograph fit QCALC is presented for the rising parameter of -3.7053 and a falling parameter of -8.1642. The actual dimensionless hydrograph is presented in the 4th column.

YEAR	MO	DAY	WTS
57	5	24	426

PEAK	TPEAK
2.2000	.4333

RISE	FALL
-3.7053	-8.1642

HOURS	T/TP	Q	Q/QP	QCALC
0.0000	0.0000	.0056	.0025	0.00000000
.0166	.0384	.0144	.0065	.00000001
.0833	.1923	.1300	.0590	.03407449
.1166	.2692	.3420	.1554	.13408484
.1666	.3845	.8300	.3772	.36848544
.2166	.4999	1.3000	.5909	.60572557
.2666	.6153	1.6800	.7636	.78885133
.3166	.7307	1.9400	.8818	.90837238
.3499	.8076	2.0800	.9454	.95714221
.3999	.9230	2.1800	.9909	.99400277
.4333	.9999	2.2000	.9999	1.00000020
.4666	1.0769	2.1700	.9863	.98898750
.4999	1.1538	2.1300	.9681	.96000113
.5499	1.2692	2.0200	.9181	.89444890
.5999	1.3846	1.8500	.8409	.81472691
.6499	1.5000	1.6500	.7499	.73051512
.7499	1.7308	1.2500	.5681	.57021832
.8166	1.8846	1.0000	.4545	.47729610
.8999	2.0769	.8240	.3745	.37951267
1.0666	2.4616	.5400	.2454	.23853391
1.1666	2.6923	.3530	.1604	.18108410
1.2333	2.8461	.2990	.1359	.15110291
1.3166	3.0385	.2680	.1218	.12095815
1.5166	3.5000	.1840	.0836	.07226657
1.7999	4.1539	.1150	.0522	.03653790
2.0666	4.7693	.0756	.0343	.02018948
2.3999	5.5385	.0509	.0231	.01020622
3.9833	9.1925	.0144	.0065	.00077044
5.1499	11.8848	.0072	.0032	.00018107
9.1333	21.0774	.0018	.0008	.00000547
9.1499	21.1158	.0018	.0008	.00000541
10.1499	23.4236	.0015	.0006	.00000277
11.1499	25.7314	.0010	.0004	.00000150
12.1499	28.0391	.0005	.0002	.00000085